

17 Der Analog/Digital-Wandler

Der Controller 80535 besitzt acht Analogeingänge. Er hat jedoch nur eine Schaltung zur Analog/Digital-Wandlung. Deshalb müssen die acht Analogeingänge über einen Multiplexer auf die Wandlerschaltung gegeben werden. Die Wandlerschaltung vergleicht die analoge Eingangsspannung mit einer selbst erzeugten Analogspannung. Während dieses Vergleiches darf sich die Eingangsspannung nicht mehr ändern. Deshalb wird die Eingangsspannung zu einem bestimmten Zeitpunkt erfasst und dann intern gehalten (Sample and Hold).


In der Sample-and-Hold-Schaltung wird im Prinzip ein Kondensator von der Eingangsspannung aufgeladen. Anschließend wird die Eingangsspannung abgetrennt und der Kondensator hält nun die Spannung für den folgenden Vergleich.

Der A/D-Wandler vergleicht die gehaltene Eingangsspannung mit einer intern eingestellten analogen Spannung. Die interne Vergleichsspannung wird aus einer Referenzspannung und einem digital geschalteten Netzwerk gebildet. Stimmt die digital gebildete Vergleichsspannung mit der Eingangsspannung überein, ist die A/D-Wandlung beendet. Die analoge Eingangsspannung ist dann gleich dem Digitalwert, der die Vergleichsspannung eingestellt hat.

Für den Vergleich wird der 8-Bit-Digitalwert zur Erzeugung der Vergleichsspannung, von null ausgehend, systematisch erhöht. Um mit möglichst wenig Einstellungen den Wert der Eingangsspannung zu erreichen, wird der Digitalwert nach dem Prinzip der sukzessiven Approximation (Wägeverfahren) verändert. Damit wird die analoge Eingangsspannung in acht Einstellschritten erreicht.

Der gesuchte Digitalwert steht nach der Wandlung im Spezial-Funktions-Register ADDAT.


Blockschaltbild der A/D-Wandler-Baugruppe:




17.1 Die Referenzspannung

Aus der Referenzspannung wird über ein digital geschaltetes Netzwerk die Vergleichsspannung für das analoge Eingangssignal erzeugt.

Die Referenzspannung wird an den Anschlüssen V_{AREF} und V_{AGND} des Controllers angeschlossen. Sie beträgt 5 V und ist nicht variabel.




Reicht die Genauigkeit aus, ist es am einfachsten, die Versorgungsspannung auch als Referenzspannung zu verwenden.



Die Referenzspannung wird über einen 8-Bit-Digitalwert in 256 Spannungswerte aufgeteilt. Die Auflösung erfolgt also in 255 Stufen.

$$5 \text{ V} : 255 = 0,01960 \text{ V} \approx 20 \text{ mV}$$



Einstellbare Referenzspannung


Um die Genauigkeit der Wandlung zu erhöhen, lässt sich der Referenz-Spannungsbereich, der in 255 Stufen geteilt wird, intern verringern.

Dazu kann V_{AGND} angehoben und V_{AREF} abgesenkt werden. Dies geschieht per Programm durch Einstellungen im Register DAPR (D/A-Converter Programming).

Mit DAPR stellt sich die interne Referenzspannung ein, aus der sich die Vergleichsspannung ableitet.

Die Genauigkeit wird größer, weil der verringerte Referenz-Spannungsbereich wieder in 255 Stufen geteilt wird.

Soll die interne Verringerung der Referenzspannung nicht genutzt werden, ist der Wert 00h in das Register DAPR zu schreiben. Mit dem Einschreiben des Wertes in das Register DAPR wird die Wandlung gestartet.



Wahl der Referenzspannung

Über je vier Bit im Register DAPR lassen sich V_{intREF} und V_{intGND} einstellen.

| DAPR | nicht bitadressierbar | | | | Resetwert 00h | | | |
|--------------|-----------------------|---|---|--------------|---------------|---|---|---|
| 7 | | | | | | | | 0 |
| V_{intREF} | | | | V_{intGND} | | | | |
| Stufe 0 = 0 | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 0 |
| 1 = 0 | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 1 |
| 14 = 1 | 1 | 1 | 0 | 0 | 1 | 1 | 1 | 0 |
| 15 = 1 | 1 | 1 | 1 | 1 | 1 | 1 | 1 | 1 |

Spannung pro Stufe:

$$\frac{5 \text{ V}}{16} = 0,3125 \text{ V}$$

Es lassen sich jedoch nicht alle Stufen nutzen. Außerdem soll die Spannungsdifferenz zwischen oberer und unterer Grenze mindestens 1 V betragen.

Tabelle der einstellbaren Spannungen:

| Stufe | V_{intREF} in V | V_{intGND} in V |
|---------|----------------------|----------------------|
| 0 | 5,0 | 0,0 |
| 1 | - | 0,3125 |
| 2 | - | 0,625 |
| 3 | - | 0,9375 |
| 4 | 1,25 | 1,25 |
| 5 | 1,5625 | 1,5625 |
| 6 | 1,875 | 1,875 |
| 7 | 2,1875 | 2,1875 |
| 8 | 2,5 | 2,5 |
| 9 | 2,8125 | 2,8125 |
| 10 = Ah | 3,125 | 3,125 |
| 11 = Bh | 3,4375 | 3,4375 |
| 12 = Ch | 3,75 | 3,75 |
| 13 = Dh | 4,0625 | - |
| 14 = Eh | 4,375 | - |
| 15 = Fh | 4,6875 | - |

Beispiel 17.1

Ein analoger Geber verändert seine Spannung von 1 V bis 3 V. Um die mögliche Genauigkeit auszunutzen, wird die interne Referenzspannung diesem Bereich angepasst.

Gewählt:


$$\begin{aligned} V_{\text{intGND}} &= 0,9375 \text{ V} & V_{\text{intREF}} &= 3,125 \text{ V} \\ &= \text{Stufe } 10 \text{ } 3 & &= \text{Stufe } 3 \text{ } 10 \\ &= A \text{ } 3 & &= 3 \text{ } A \end{aligned}$$

In Register DAPR ist die Hexzahl A3 zu schreiben. `MOV DAPR, #A3h`


Damit ist der Wandelbereich festgelegt und gleichzeitig wird die Wandlung gestartet.

Jetzt wird der Bereich von 0,9375 bis 3,125 in 255 Stufen aufgeteilt.


$$\frac{3,125 \text{ V} - 0,9375 \text{ V}}{255} = 8,57 \text{ mV}$$

**17.2 Analogeingänge**

Die analogen Eingangsspannungen dürfen die Referenzspannung nicht überschreiten.



Einer der acht Analogeingänge wird über einen Multiplexer auf den A/D-Wandler geschaltet. Welcher Analogeingang gewandelt werden soll, lässt sich über die unteren drei Bits im Register ADCON (A/D-Converter) wählen.



Befehle zur Wahl von Analogeingang fünf:


`ANL ADCON, #F8h`
`ORL ADCON, #05h`

17.3 Betriebsarten

Der Wandler lässt sich in zwei Arten betreiben:

1. Betriebsart einmalige Wandlung,
2. Betriebsart ständige Wiederholung.

Die Betriebsart lässt sich mit Bit ADM (A/D-Converter-Modus) in Register ADCON einstellen.

**17.4 Wandelzeiten**


Die Wandelzeiten beziehen sich auf die Zykluszeit des Controllers, die wiederum von der Quarzfrequenz abhängt. Ein Zyklus besteht aus 12 Takten.

Die Wandelzeit lässt sich in drei Zeiten unterteilen.

 $t_L = \text{Load-time}$ $t_s = \text{Sample-Time}$ t_C = Conversion-Time

Der Spannungswert, der am Ende der Load-Time erreicht ist, wird in den Digitalwert gewandelt. Nach der Conversion-Time steht der Digitalwert in Register ADDAT zur Verfügung.

Diagramm der Wandelzeiten:



Die Dauer der Wandlung wird durch das Busy-Bit BSY in Register ADCON angezeigt. Es wird durch die interne Hardware zu Beginn der Wandlung gesetzt und am Ende zurückgesetzt. Dieses Busy-Bit kann vom Programm abgefragt werden. Außer dem Busy-Bit wird nach beendeter Wandlung ein Interrupt-Bit IADC in Register IRCON gesetzt. Dieses Interrupt-Bit muss per Programm zurückgesetzt werden.

Datenblatt-Angaben:

A/D Converter Characteristics

$$V_{CC} = 5 \text{ V} \pm 10 \%;$$

$$V_{SS} = 0 \text{ V};$$

$$V_{\text{AREF}} = V_{\text{CC}} \pm 5\%;$$

$$V_{AGND} = V_{SS} \pm 0,2 \%;$$

$$V_{\text{IntAREF}} - V_{\text{IntAGND}} \geq 1 \text{ V};$$

Controller 80535

| Parameter | Symbol | Limit values | | | Unit | Test condition |
|--------------------------------------------|-----------------------|-------------------------|------|-------------------------|------|---------------------------------------------------------------------------------------------------------------------------------|
| | | min. | typ. | max. | | |
| Analog input voltage | V_{INPUT} | $V_{\text{AGND}} - 0.2$ | — | $V_{\text{AREF}} + 0.2$ | V | — |
| Analog input capacitance | C_{I} | — | 25 | — | pF | 1) |
| Load time | t_{L} | — | — | $2t_{\text{CY}}$ | μs | — |
| Sample time (incl. load time) | t_{S} | — | — | $5t_{\text{CY}}$ | μs | — |
| Conversion time (including sample time) | t_{C} | — | — | $15t_{\text{CY}}$ | μs | — |
| Differenzial non-linearity | DNLE | — | ±1/2 | ±1 | LSB | $V_{\text{INAREF}} =$ $V_{\text{AREF}} = V_{\text{CC}}$ $V_{\text{INTAGND}} =$ $V_{\text{AGND}} = V_{\text{SS}}$ 2) |
| Integral non-linearity | INLE | — | ±1/2 | ±1 | LSB | |
| Offset error | — | — | ±1/2 | ±1 | LSB | |
| Gain error | — | — | ±1/2 | ±1 | LSB | |
| Total unadjusted error | TUE | — | ±1 | ±2 | LSB | |
| V_{AREF} supply current | I_{REF} | — | — | 5 | mA | 2) |
| Internal reference error | V_{INTREFER} | — | ±5 | ±15 | mV | 2) |

1) The internal resistance of the analog source must be low enough to assure full loading of the sample capacitance (C_1) during load time (t_L). After charging of the internal capacitance (C_1) in the load time (t_L) the analog input must be held constant for the rest of the sample time (t_S).

2) The differential impedance (r_D) of the analog reference voltage source must be less than 1 k Ω at reference supply voltage.

Controller 80C535

| Parameter | Symbol | Limit values | | | Unit | Test condition |
|--------------------------------------------|----------------|------------------|------|------------------|------|---------------------|
| | | min. | typ. | max. | | |
| Analog input voltage | V_{AINPUT} | $V_{AGND} - 0.2$ | — | $V_{AREF} + 0.2$ | V | 5) |
| Analog input capacitance | C_I | — | 25 | 45 | pF | 3) |
| Load time | t_L | — | — | $2t_{CY}$ | μs | — |
| Sample time (incl. load time) | t_S | — | — | $7t_{CY}$ | μs | — |
| Conversion time (including sample time) | t_C | — | — | $13t_{CY}$ | μs | — |
| Differenzial non-linearity | DNLE | — | ±1/2 | ±1 | LSB | $V_{INTAREF} =$ |
| Integral non-linearity | INLE | — | ±1/2 | ±1 | LSB | $V_{AREF} = V_{CC}$ |
| Offset error | — | — | ±1/2 | ±1 | LSB | $V_{INTAGND} =$ |
| Gain error | — | — | ±1/2 | ±1 | LSB | $V_{AGND} = V_{SS}$ |
| Total unadjusted error | TUE | — | ±1 | ±2 | LSB | 3) |
| V_{AREF} supply current | I_{REF} | — | — | 5 | mA | 4) |
| Internal reference error | $V_{INTREFER}$ | — | — | ±30 | mV | 4) |

3) The output impedance of the analog source must be low enough to assure full loading of the sample Capacitance (C_i) during load time (t_L). After charging of the internal capacitance (C_i) in the load time (t_L) the analog input must be held constant for the rest of the sample time (t_s)

5) Exceeding these limit values at one or more input channels will cause the device to shut down.

5) Exceeding these limit values at one or more input channels will cause additional current which is sinked/sourced of these channels. This may also affect the accuracy of other channels which are operated within these specifications.

Beispiel 17.2

Die analoge Spannung am Eingang AN2 ist einzulesen und in eine 8-Bit-Digitalzahl zu wandeln. Die Spannung variiert im Bereich von 0 bis 5 V.

Nach Aufruf des Unterprogramms AD1_2 soll der Kanal 2 einmalig gewandelt werden.

```
AD1_2:  CLR ADM           ;Betriebsart einmalige Wandlung
        ANL ADCON, #0F8h ;Wahl des Analogeingangs AN2
        ORL ADCON, #02h
        MOV DAPR, #00h   ;Referenzspannung 0 bis 5 V und Start
        JB BSY, WASTE    ;warten bis Wandlung Ende
        RET
```

Übung 17.1

Schreiben sie ein Unterprogramm AD1_1 zur einmaligen Wandlung von Kanal 1. Die Analogspannung am Eingang kann maximal 3 V betragen. Arbeiten sie mit der maximalen Auflösung für diesen Bereich.


In ADDAT steht nach der Wandlung z. B. die Hexzahl D3. Welche Eingangsspannung lag an?

17.5 Anwendung als Messwert-Aufnehmer

Es sollen acht analoge Messwerte aufgenommen und gespeichert werden. Die Messwerte liegen als in der Regelungstechnik übliche normierte Spannungen von 0 bis 10 V vor. Sie sind über einen Spannungsteiler und eine Schutzbeschaltung auf 5 V herabzusetzen. Die Messwertaufnahme ist mit einer positiven Flanke an einem digitalen Porteingang zu starten (Signal: „Messung Start“).

Die gemessene Eingangsspannung ist über eine Siebensegment-Anzeige als Spannung von 0,0 V bis 9,9 V anzuzeigen. Da für die acht Messwerte nur eine einzige zweistellige Anzeige vorhanden ist, soll sich der anzuzeigende Messwert über einen Wahlschalter an einem digitalen Eingangsport einstellen lassen.

Controller mit Eingangs- und Ausgangsbeschaltung:



Bei der Lösung soll, ausgehend vom Wandlerprogramm für einen Analogwert, das Wandlerprogramm für acht Analogwerte entwickelt werden. Um dieses Wandlerprogramm für acht Analogwerte wird dann das Hauptprogramm mit Start-Erkennung aufgebaut, Wahl des anzuzeigenden Messwertes und Ausgabe der normierten Spannung im BCD-Code.

A/D-Wandler für 1 Analogwert

Welcher der acht Analogwerte gewandelt werden soll, steht in der Adresse AKANAL. AKANAL muss vor dem Aufruf des Wandlerprogramms AD1 mit dem Wert 0 bis 7 geladen werden. Das Ergebnis der Wandlung steht in ADDAT.

```
AD1:  CLR ADM           ;Betriebsart einmalige Wandlung
        ANL ADCON, #0F8h ;einsetzen des Analogkanals
        MOV A, AKANAL
        ORL ADCON, A
        MOV DAPR, #00h   ;Referenzspannung 0 bis 5 V und Start
        JB BSY, A11      ;warten auf Wandlung Ende
        RET
```


A/D-Wandlung für maximal 8 Analogwerte

Das erste Wandelergebnis wird zu der Adresse gebracht, die in AD_ANF steht. Die nächsten Ergebnisse werden auf den folgenden Adressen abgespeichert. AD_ANF muss vor Aufruf des Wandlerprogramms AD8 mit der gewünschten Adresse geladen werden. Die Wandlung beginnt bei dem Analogeingang, der in Adresse AKANAL steht und endet bei dem Analogeingang, der in Adresse EKANAL steht. Die Werte für AKANAL (z. B. 00) und EKANAL (z. B. 07) müssen vor Aufruf von AD8 in die Adressen geladen werden.

Assemblerprogramm:

```

;*****
;Einlesen und Abspeichern von maximal 8 Analogwerten
;*****
Title "A/D-Wandlung von max 8 Analogwerten, Datei ADW8.ASM"
controller 80535
;
;----- Unterprogramm 8-Kanal A/D-Wandler -----
;Das erste Wandelergebnis wird zu der Adresse gebracht, die in
;Adresse AD_ANF steht. Die nächsten Ergebnisse gehen in die folgenden
;Adressen.
;Die Wandlung beginnt bei dem Kanal, dessen Nummer in Adresse AKANAL
;steht und endet bei dem Kanal, der in Adresse EKANAL steht.
;
ad8: mov r0,AD_ANF          ;Ergebniszeiger r0 auf AD_ANF stellen
      mov kanal,akanal      ;Nummer des ersten zu wandelnden Kanals
ad81: lcall ad1              ;1 Analogwert wandeln
      mov @r0,addat         ;Ergebnis nach Adresse in AD_ANF + n
      inc r0                ;Ergebniszeiger + 1
      mov a,kanal
      inc kanal              ;Analogkanal + 1
      cjne a,ekanal,ad81     ;letzter Analogkanal erreicht? Nein:
      ret                   ;Ja:
;
;----- Unterprogramm 1-Kanal A/D-Wandler -----
;Der zu wandelnde Analogkanal muss in Adresse KANAL stehen.
;Das Wandelergebnis steht in SFR-Register ADDAT.
;
ad1:  clr adm                ;Betriebsart einmalige Wandlung
      anl adcon,#0f8h        ;Einsetzen des Analogkanals
      mov a,kanal             ;= letzte 3 Bit von ADCON
      orl adcon,a
      mov dapr,#00h          ;Referenzspannung 0 bis 5V und Start
ad11: jb bsy,ad11            ;warten auf Ende der Wandlung
      ret
;*****

```

Hauptprogramm A/D-Wandler

Das Hauptprogramm soll zyklisch bearbeitet werden. Die Eingangssignale des digitalen Ports werden zu Programmbeginn eingelesen und abgespeichert. Dann erfolgt die Abfrage, ob an „Start“ eine positive Flanke angelegen hat. Ist das der Fall, wird AD8 aufgerufen. Zur Erkennung

der positiven Flanke wird das „Start“-Signal des vorhergehenden Programmzyklus, welches unter START_A abgespeichert wurde, mit dem neu eingelesenen „Start“-Signal verglichen. Ist das alte „Start“ gleich 0 und das neue 1, liegt eine positive Flanke vor.

```

ADW1:  MOV EIN1,P1           ;einlesen und abspeichern der
                                   ;Eingangssignale P1
                                   JNB START,ADW2       ;positive Flanke an START? Nein:
                                   JE START_A,ADW2       ;Nein:
                                   LCALL AD8              ;Ja: Wandlerprogramm aufrufen
ADW2:  MOV C,START           ;"Start" Neu nach "Start" Alt
                                   MOV START_A,C

```

Die gewandelten Analogwerte stehen ab Adresse AD_ANF an aufwärts. Sie müssen für die Anzeige skaliert und in eine zweistellige BCD-Zahl umgeformt werden.

Skalieren

| U _{Geber} | U _{Controller} | Dualzahl | Anzeigewert |
|--------------------|-------------------------|----------|-------------|
| 0 V | 0,0 V | 0 | 0,0 V |
| 2 V | 1,0 V | 51 | 2,0 V |
| 3 V | 1,5 V | 76 | 3,0 V |
| 4 V | 2,0 V | 102 | 4,0 V |
| 5 V | 2,5 V | 127 | 5,0 V |
| 6 V | 3,0 V | 153 | 6,0 V |
| 7 V | 3,5 V | 178 | 7,0 V |
| 8 V | 4,0 V | 204 | 8,0 V |
| 9 V | 4,5 V | 229 | 9,0 V |

Ein Vergleich der Zahlen zeigt, dass sich der Anzeigewert nach nebenstehender Gleichung berechnen lässt.

$$\text{Anzeigewert} = \frac{\text{Dualzahl}}{5} \times 2$$

nur ganze Zahl

Um die Skalierung einfach zu halten, soll nur mit Zahlen von einem Byte Länge gearbeitet werden. Das bedingt, dass die Dualzahl zuerst durch fünf geteilt werden muss. Dabei fallen die Stellen hinter dem Komma weg. Anschließend wird mit zwei multipliziert. Diese Multiplikation lässt sich am einfachsten durch eine Verschiebung um eine Stelle nach links realisieren.

Codewandler Dual in BCD

Der berechnete Anzeigewert liegt als Dualzahl vor und muss zur Anzeige in eine zweistellige BCD-Zahl umgeformt werden. Diese Codewandlung lässt sich einfach realisieren, da bei einer Division durch zehn die ganze Zahl im Akku und der Rest in Register B steht. Die ganze Zahl sind die Zehner, der Rest die Einer. Einer und Zehner müssen dann noch in ein Byte gepackt werden.

Assemblerprogramm:

```

;*****
;Aufnahme und Anzeige von maximal 8 analogen Messwerten 0 bis 10V
;*****
Title "A/D-Wandler, Datei ADWAN1.ASM"
controller 80535
;
;----- Zuordnungen -----
;
org 8000h
akanal equ 30h      ;Adresse für die Nummer des ersten zu wandelnden
                    ;Analog-Kanals
kanal equ 31h        ;Adresse für die Nummer des momentanen Kanals
ekanal equ 32h       ;Adresse für die Nummer des letzten zu wandelnden
                    ;Analog-Kanals
ad_anf equ 33h       ;Adresse für die Adresse des ersten
                    ;Wandlerergebnisses
ein1 equ 20h          ;Eingangssignal-Abbild
aus5 equ 21h          ;Ausgangssignalabbild
merk equ 22h          ;Speicher für interne Bitvariable
start equ ein1.0      ;Taste "Messung Start"
start_alt equ merk.0  ;Start Alt (aus vorhergehendem Zyklus)
;
;----- Initialisierung -----
;
mov akanal,#00h      ;Wandlung von Kanal 0 bis 7
mov ekanal,#07h
mov ad_anf,#90h      ;Wandlerergebnis Kanal 0 nach Adresse 90h
mov ein1,#00h
mov aus5,#00h
mov p1,0ffh          ;P1 vorbereiten für Eingabe
;
;----- Programm -----
;
;Erfassen der positiven Flanke an START, um die Wandlung zu starten:
adw1:mov ein1,p1      ;einlesen
      jnb start,adw2  ;positive Flanke an START? Nein:
      jb start_alt,adw2 ;Nein:
      lcall ad8        ;Ja: A/D-Wandlung Ein
adw2:mov c,start      ;neues START-Signal nach START_Alt
      mov start_alt,c
;
;Skalieren und Anzeigen des Messwertes, der am Wahlschalter eingestellt
;ist:
mov a,ein1            ;gewählter Analogeingang nach A
anl a,#11110000b
swap a               ;rechtsbündig schieben
rr a
add a,ad_anf         ;Messwertadresse ermitteln
mov r0,a             ;Messwertadresse steht in R0
mov a,@r0            ;gewählter Messwert nach A

```

```

mov b,#05h           ;skalieren
div ab
rl a                 ;Anzeigewert steht als Dualzahl in A
mov b,#10            ;Codewandeln in BCD
div ab
swap a
orl a,b              ;Anzeigewert steht als BCD in A
mov aus5,a
mov p5,aus5          ;Ausgabe Anzeigewert
ljmp adw1            ;zyklische Bearbeitung
;*****
include "adw8.asm"
;*****
;*****
;Einlesen und Abspeichern von maximal 8 Analogwerten
;*****
Title "A/D-Wandlung von max 8 Analogwerten, Datei ADW8.ASM"
controller 80535
;
;----- Unterprogramm 8-Kanal A/D-Wandler -----
;
;Das erste Wandlerergebnis wird zu der Adresse gebracht, die in
;Adresse AD_ANF steht. Die nächsten Ergebnisse gehen in die folgenden
;Adressen.
;Die Wandlung beginnt bei dem Kanal, dessen Nummer in Adresse AKANAL
;steht und endet bei dem Kanal, der in Adresse EKANAL steht.
;
ad8:  mov r0,#90h      ;Ergebniszeiger r0 auf AD_ANF stellen
      mov kanal,akanal ;Nummer des ersten zu wandelnden Kanals
ad81: lcall ad1        ;1 Analogwert wandeln
      mov @r0,addat    ;Ergebnis nach Adresse in AD_ANF + n
      inc r0           ;Ergebniszeiger + 1
      mov a,kanal      ;Analogkanal + 1
      inc kanal        ;Analogkanal + 1
      cjne a,ekanal,ad81 ;letzter Analogkanal erreicht? Nein:
      ret              ;Ja:
;
;----- Unterprogramm 1-Kanal A/D-Wandler -----
;
;Der zu wandelnde Analogkanal muss in Adresse KANAL stehen.
;Das Wandlsergebnis steht in SFR-Register ADDAT.
;
ad1:  clr adm          ;Betriebsart einmalige Wandlung
      anl adcon,#0f8h  ;Einsetzen des Analogkanals
      mov a,kanal      ;:= letzte 3 Bit von ADCON
      orl adcon,a
      mov dapr,#00h    ;Referenzspannung 0 bis 5V und Start
ad11: jb bsy,ad11      ;warten auf Ende der Wandlung
      ret
;*****

```